

Embolica que fa fort (4 punts)

Donat el següent codi:

```
PROGRAMA PRINCIPAL
VAR x, y, z: INTEGER;
    n: ARRAY [0..9, 0..9] OF CHAR;
    m: ARRAY [0..9] OF CHAR;

FUNCTION B (VAR matriu: array[0..9, 0..9] OF CHAR;
            fila_i: INTEGER;
            VAR col_i: INTEGER;): INTEGER;
BEGIN
    B:=matriu[filai - 1, col_i]; (1)
END (*final de B*)

BEGIN
...
x:=B(n, y, z); (2)
...
END (*final de principal*)
```

Consideracions:

- Les matrius estaran emmagatzemades en posicions **no** consecutives de memòria i per files.
- Els paràmetres s'empilen de dreta a esquerra.
- Una variable de tipus integer ocupa un longword i una de tipus char ocupa un byte.

Es demana:

- a) Codificar les estructures de dades del programa principal en L.E.
- b) Codificar com es crida la subrutina B en L.E., o dit d'una altra manera, codificar en L.E. l'instrucció (2).
- c) Escriure en L.E. la subrutina B.
- d) Dibuixar l'estat de la pila en el moment d'executar l'instrucció (1).
- e) Fer un programa en L.E. que emplei la taula m amb els elements de la columna COL de la matriu n. Aquest programa ha de cridar la subrutina B. Suposeu que COL és una constant.

A on salta en Wally? (2 punts)

Donat el següent codi:

```
SUB1:   ADDL (R1), (R2)
        RSB
SUB2:   ADDL (R2), (R1)
        RSB
W:      .LONG 1, 2, 3, 4, 5, 6
        .ENTRY INICI, 0
        CLRL R3
        CLRL R0
        MOVL #W, R4
BUCLE:  MOVAL (R4)+, R1
        MOVAL (R4)+, R2
        MOVL #CONT, -(SP)
WALLY:  JMP (R3)
CONT:   ADDL #4, R3
        INCL R0
        CMLL R0, #2
        BNEQ BUCLE
        $EXIT S
        .END INICI
```

Nota:

- Suposem que quan executeu aquest programa, el valor assignat a l'etiqueta SUB1 serà 0.

Es demana:

1. Tradueix a codi màquina les instruccions de la subrutina SUB1.
2. Explica quants accessos a memòria es produeixen quan s'executen les dues instruccions de la subrutina SUB1.
3. Quin mètode s'utilitza per a passar els paràmetres a les subrutines. Comentar també si els paràmetres es passen per valor o per referència. Com es retorna el resultat?
4. Indica quin serà el contingut del vector W després d'executar-se el codi.

@	(@) Abans	(@) Després
W	1	
W+4	2	
W+8	3	
W+12	4	
W+16	5	
W+20	6	

Solució Embollica que fa fort

a) estructures de dades:

```
x: .BLKL 1
y: .BLKL 1
z: .BLKL 1
n: .LONG fila0,filal,...,fila9
fila0: .BLKB 10
filal: .BLKB 10
...
fila9: .BLKB 10
m: .BLKB 10
```

b) cridar a la subrutina B:

```
MOVL #z,-(SP)
MOVL y,-(SP)
MOVL #n,-(SP)
CLRL -(SP);reservem espai pel resultat
BSBW B
MOVL (SP)+,x
ADDL #12,SP
```

c) la subrutina B en L.E.:

```
B: PUSHR #^M<R0,R1,R2>
MOVL 12+0(SP),R0;R0=@ de n
MOVL 12+12(SP),R1;R1=fila i
MOVL 12+16(SP),R2;R2=@ de z
MOVL (R2),R2;R2=z
DECL R1;R1=fila_i-1
MOVL (R0)[R1],R0;R0=@ de fila_i-1
MOVB (R0)[R2],12+4(SP)
POPR #^M<R0,R1,R2>
RSB
```

d) estat de la pila en (1):

SP:	R0
SP+4:	R1
SP+8:	R2
SP+12:	@ de retorn
SP+16:	0
SP+20:	@ de n
SP+24:	y
SP+28:	@ de z

e) emplenar la taula m:

```
MOVL #1,R0
MOVL #COL,z
MOVL #z,-(SP)
MOVL R0,-(SP)
MOVL #n,-(SP)
CLRL -(SP)
bucle: BSBW B
DECL R0
MOVB (SP),m[R0]
ADDL #2,R0
MOVL R0,8(SP)
CML R0,#10
BLEQ bucle
ADDL #16,SP
```

Solució A on salta en Wally ?

1. traduir a L.M. SUB1:

```
62 61 C0      0000      ADDL (R1), (R2)
           05      0003      RSB
```

2. accessos a memòria al executar SUB1:

instrucció	accés	motiu
ADDL (R1), (R2)	Lectura d'1 byte	fetch del C.O.
	Lectura d'1 byte	fetch del E. del 1er Operand
	Lectura d'1 long	llegir el 1er operand
	Lectura d'1 byte	fetch del E. del 2on Operand
	Lectura d'1 long	llegir el 2on operand
	Escriure 1 long	escriure el resultat de la suma
RSB	Lectura d'1 byte	fetch del C.O.
	Lectura d'1 long	llegir l'@ de retorn

3. Quin mètode s'utilitza per a passar els paràmetres a les subrutines. Comentar també si els paràmetres es passen per valor o per referència. Com es retorna el resultat ?

- Els paràmetres es passen a través dels registres i per referència, ja que el que es passa són les adreces dels números que la subrutina tractarà. Per tant la subrutina modificarà les variables del programa principal. El resultat no es torna per la pila ni per cap registre, sino que el resultat d'executar la subrutina és que es modifiquen variables del programa principal.

4. el contingut del vector W després d'executar-se el codi:

@	(@) Abans	(@) Després
W	1	1
W+4	2	3
W+8	3	7
W+12	4	4
W+16	5	5
W+20	6	6